Advanced Programming 2017

Haskell, Continued

Andrzej Filinski
andrzej@di.ku.dk

Department of Computer Science
University of Copenhagen

September 7, 2017

Today’s topics

- Introduction to some more advanced, Haskell-specific features
  - Modules
  - Type classes
  - Laziness
  - Functional I/O principles
  - List comprehensions
- Useful to know about in own right.
- Provide important background for **monads**, next time.

Haskell's module system

- Relatively simple, compared to, e.g., Standard ML or OCaml
- But quite sufficient for most practical purposes
  - Especially in conjunction with type classes
    - Cover many (but not all) uses of ML's *parameterized* modules
- Two main purposes:
  - Namespace management
    - Using same name for unrelated purposes at different points in big program
  - Abstraction management
    - Preventing unwanted exposure of implmentation details
- Fundamental concepts: *imports* and *exports*.

Standard modules

- All Haskell code is type-checked and executed in context of some existing definitions of types and values.
- Most common definitions always visible: "standard prelude".
  - Saw several examples last time: `pi`, `(+)`, `map`, `[]`, `Maybe`, ...
- Large standard library of further functionality available:
  - *Utility* functions and data structures:
    - E.g., formatting, parsing, finite-set operations, ...
    - Could in principle by reimplmented by ordinary programmer.
    - But probably not as competently: don't re-invent the wheel!
  - System interface and control functions:
    - E.g., directory listing, exception handling, ...
    - Implementation relies on special support from compiler and/or runtime system.
    - No way to re-implement from scratch in pure Haskell code.
  - Grouped into *modules*.

Importing from modules

- To use all or parts of a module, must explicitly *import* from it.
- "`import ...`" declaration(s) must be at very beginning of file.
- Bulk import:
  - ```
    import System.Directory
    ```
  - Makes everything from module available.
  - Names may clash with own definitions, or other imports.
    - Only get error on attempted *use* of ambiguous name.
  - Normally used for "framework" modules, such as parser combinators
- Selective import
  - ```
    import System.Directory
      (getCurrentDirectory, doesFileExist)
    ```
  - Only makes explicitly listed names available.
  - Remember to enclose any operator names in parentheses.
  - Normally preferred if only need a few, unrelated functions from module in question.

Importing from modules, continued

- Qualified import
  - `import qualified Data.Set as S`
  - Like a bulk import, but prefixes all imported names with `S.`
    - `S.map :: Ord b => (a -> b) -> S.Set a -> S.Set b`
  - Avoids clash with (list-based) `map` from standard prelude
- Warning: top-level interactive loop is a bit special.
  - Can refer directly to names from other modules:

    ```
    > System.Directory.getCurrentDirectory
    "/home/andrzej/teaching/ap2017/"
    ```
  - Will not work in file; need explicit import first.
  - Prompt in top-level loop indicates which modules have been imported.
    - Can add or remove with `:mod [+/-]` *`ModName`*

Creating your own modules

- Start file containing related definitions with
  `module` *ModName* `(`*exports*`) where` *defs*
- *ModName* is the name of the module.
  - Should be the same as source filename (without trailing `.hs`).
- *exports* is comma-separated list of names (types and/or values) to be made available to users (clients) of the module.
  - Often more readable to list one name per line.
  - Use *`TypeName`*`(..)` to export a datatype together with all its constructors.
- *defs* should start with any neeed `import` declarations, as usual.

What to export from a module?

- Not specific to Haskell; general principles for API design.
- Export orthogonal set of functions useful to clients, not any internal “helper” functions you used to define them.
  - If you cannot concisely summarize what a function does, it shouldn’t be exported.
  - Arguably, it probably shouldn’t even have been defined in the first place...
    - Unclear and/or complex specifications for internal functions are a magnet for bugs.
    - Do try to formulate specification (including meanings of all parameters!) in a comment; forces you to consider what the function *should* be doing.

Example of API considerations

- Suppose we are defining a module for integer-set operations, with exports:

  ```
  empty :: IntSet
  singleton :: Int -> IntSet
  union :: IntSet -> IntSet -> IntSet
  isElt :: Int -> IntSet -> Bool
  ```

- For implementing `union`, may also have defined:

  ```
  addElt :: Int -> IntSet -> IntSet
  ```

  Should it be exported?

- Client could themselves define equivalent function:

  ```
  myAddElt x s = singleton x `union` s
  ```

  Should be almost as efficient, uses just fundamental operations.
  - If `myAddElt` significantly slower than `addElt`, maybe should improve performance of `union` in general.
  - E.g, always add elts of smaller set to larger, not vice versa.

Preventing leakage of implementation details

- Suppose we implement `IntSet` as unsorted, duplicate-free lists.
- Could just make definition in module:

  ```
  type IntSet = [Int]
  ```

  But that exposes to clients that an `IntSet` is actually a list.
  - In particular, this could evaluate to `False`:

    ```
    singleton 3 `union` singleton 4  ==
      singleton 4 `union` singleton 3
    ```
- Solution: in implementation, define a *new* type, equivalent to `[Int]`.

  ```
  newtype IntSet = IS {unIS :: [Int]}
  ```
  - Almost same as `data` with a single constructor.
  - Note: did *not* include `deriving Eq` in definition!
- Export type `IntSet`, but *not* constructor `IS`, nor projection `unIS`
  - Only use internally in module, to define `empty`, `union`, etc.
- Clients can neither create new `IntSet` values, nor inspect existing ones, except through exported API functions.
  - But then, API should probably also include an equality test.

Overloading in Haskell

- Have already seen (sometimes implicit) examples of restricted polymorphic functions:

```haskell
(==) :: Eq a => a -> a -> Bool
(+) :: Num a => a -> a -> a
show :: Show a => a -> String
```

- Haskell's type inferencer automatically keeps track of restrictions:

```
> let twice x = x + x
> :t twice
twice :: Num a => a -> a
```

- In general, may have multiple constraints:

```haskell
foo :: (Num a, Show a) => a -> String
foo x = show (x + x)
```

- Capture a uniform notion of *overloading*, where computation to be performed depends materially on types of operands and/or result.

Type classes

- A Haskell *type class* is an (open-ended) collection of types supporting a fixed set of operations.
  - Not entirely unlike *interfaces* in Java.
- Declared with `class` *`ClassName typevar`* `where` *`decls`*
  - As usual, the *decls* should align horizontally.
- Several predefined classes, including (slightly simplified):

```
class Show a where
  show :: a -> String

class Eq a where
  (==), (/=) :: a -> a -> Bool

class Num a where
  (+), (-), (*) :: a -> a -> a
  fromInteger :: Integer -> a
```

- Use `:info` *`ClassName`* in GHCi to see full list of operations.

Declaring class membership

- To include a (new or previously defined) type in a class, must add an *instance declaration*.
- Simply need to supply all the required operations of the class.
- Example (of course, better version exists in standard library):

```
data Complex = Complex {re, im :: Double}

instance Num Complex where
  (Complex r1 i1) + (Complex r2 i2) = Complex (r1+r2) (i1+i2)
  ...
  fromInteger n = Complex (fromInteger n) 0.0
```

  - **Note:** The `fromInteger n` on the RHS is *not* a recursive call, but an invocation of `fromInteger :: Integer -> Double` !
- Likewise,

```
instance Show Complex where
  show c = show (re c) ++ "+" ++ show (im c) ++ "i"
```

Numeric types in Haskell

- Actually, whole hierarchy of numeric type classes
  - `Num a`, for types `a` that have operations `(+)`, `(-)`, `(*)`
    - Mathematically: $\sim$ *rings*
  - `Fractional a`, for types `a` that also have `(/)`
    - Mathematically: $\sim$ *fields*
  - `Integral a`, for types `a` that have `div`, `mod`
    - instances: `Int`, `Integer`, ...
  - ...
- Main oddity: even *literals* are overloaded!
  - Plain 42 actually behaves like `fromInteger (42::Integer)`,
- Therefore:
  - **OK:** `pi + 1`
  - **Not OK:** `pi + length "x"`
  - **OK:** `pi + fromIntegral $ length "x"`
    - Aside: `$` often useful to avoid deeply nested parentheses
    - Just a right-associative infix application operator.

More type-class details

- Class inheritance
  - Can also constrain type variable in class declaration

    ```
    class Bar a => Foo a where ...
    ```

  - Can only declare a type to be instance of `Foo`, if it's already an instance of `Bar`.
  - Ex: `class Eq a => Ord a where (<) :: a -> a -> Bool; ...`
- Default implementations
  - Can include a *default* definition of a class operation:

    ```
    class Eq a where
      (==), (/=) :: a -> a -> Bool
      x /= y = not (x == y)
    ```

  - In instance declaration, if we omit definition for `(/=)`, the default one is used
  - Note: default implementation may use operations of superclass.
- Both features a bit esoteric, but recent API change for `Monad` class makes them unavoidable...

Automatically deriving instances

- Haskell can automatically construct *certain* instance declarations for newly defined types.
  - Only for a few built-in classes (need compiler support)
- `data MyType = ...  deriving (Eq, Show, Read, ...)`
- Derived `Show`:
  - Displays values in a format parseable as source code.
  - E.g., `"Complex {re = 3.0, im = 4.2}"`
  - Whereas our custom `show` would return `"3.0+4.2i"`
- Derived `Eq`:
  - Structural equality (assuming all constituent types have `Eq` instances.
  - Usually fine, but sometimes want a coarser notion of equality.
  - E.g., in our module implementing `IntSet`:

```haskell
instance Eq IntSet where
  (IS xs) == (IS ys) = all (\x -> x `elem` ys) xs &&
                       all (\y -> y `elem` xs) ys
```

Monoids

- Another common class: types with notion of "accumulation"

```
class Monoid a where
  mempty :: a
  mappend :: a -> a -> a

instance Monoid String where
  mempty = "" ; mappend = (++)

instance Monoid Int where
  mempty = 0 ; mappend = (+)

instance (Monoid a, Monoid b) => Monoid (a,b) where
  mempty = (mempty {-of type a-}, mempty {-of type b-})
  mappend (a1,b1) (a2,b2) = (mappend a1 a2, mappend b1 b2)
```

- All `Monoid` instances *a* *should* satisfy, for all $x, y, z :: a$

```
mappend mempty x == x,  mappend x mempty == x,
mappend x (mappend y z) == mappend (mappend x y) z
```

Constructor classes

- Can also classify *type constructors* (parameterized types).
- Example: *functors*, for “container-like” type constructors

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b

instance Functor [] where  -- type [a] stands for [] a
  fmap = map

data Tree a = Leaf a | Node (Tree a) (Tree a)
instance Functor Tree where
  fmap f (Leaf a) = Leaf (f a)
  fmap f (Node tl tr) = Node (fmap f tl) (fmap f tr)
```

  Then, `fmap odd $ Node (Leaf 2) (Node (Leaf 3) (Leaf 5))` evaluates to `Node (Leaf False) (Node (Leaf True) (Leaf True))`

- All `Functor` instances *should* satisfy:
  `fmap id == id`, `fmap (g . f) == fmap g . fmap f`.

Laziness

- Unlike most languages, Haskell has a *lazy* ($\sim$ *non-strict*) semantics.
- Subexpressions not evaluated until their values actually needed.
- To illustrate behavior, `undefined` is a predefined expression that causes a runtime error when evaluated.
- Sample interaction:

```
> let x = undefined in x + 1
*** Exception: Prelude.undefined
> let x = undefined in 3
3
```

- Even if everything terminates (eventually), lazy evaluation may avoid wasting work: `let x =` *`bigExp`* `in 0`
- But in `let x =` *`bigExp`* `in x+x`, Haskell will *memoize* ($\approx$ *cache*) value of `x` after first use, to avoid recomputation.
  - Only safe because *`bigExp`* cannot have side effects!
- Same behavior for function arguments ("call-by-need")

```haskell
let f x = 42 in f undefined  -- immediately returns 42
```

Lazy evaluation, continued

- Even when result of subexpression is used, it will only be evaluated enough to allow computation to proceed:

```
let p = (undefined, 3) in snd p  --  returns 3

case Just undefined of
  Nothing -> False ; Just x -> True  -- returns True
```

- In general, evaluation of all constructor arguments (including tuples and list nodes, but *not* `newtype`) is delayed.
  - Can inadvertently construct “booby-trapped” values that only explode when accessed.
  - Commonly: only when being printed as results.

    ```
    > let l = [10,20,undefined,40] in (length l, show l)
    (4,"[10,20,*** Exception: Prelude.undefined
    ```

  - The top-level printer is *forcing* evaluation.
  - Apocryphal lecture by Simon Peyton Jones (GHC developer): “This is a talk about lazy evaluation. Are there any questions?”

Streams

- In most practical situations, lazy vs. eager evaluation of functional program makes no difference.
  - Rare to write a subexpression, then never use its result (dead code)
- But lazy evaluation makes it particularly simple and natural to work with *infinite* lists (*streams*).
- Just like functions can be recursively defined, so can list values:

```
ones, nats :: [Int]
ones = 1 : ones
nats = 0 : map (\x -> x+1) nats
```

- `> take 5 nats` prints `[0,1,2,3,4]`
- `> drop 5 nats` prints `[5,6,7,8,9,10,11,`... until interrupted.
- Again, the top-level printer drives the actual computation.

Introduction to Haskell I/O

- Haskell is a completely pure language, no side effects allowed.
- So how can we possibly write Haskell programs that interact with the real world?
  - File system, terminal, network, other OS services,....
- Answer: top-level printer itself doesn't need to be pure!
- Can have pure program compute a lazy list (stream) of I/O requests (actions) for top-level printer to perform.
  - *Producing* the list itself is effect-free; *obeying* it is not.
  - The list is inspected incrementally, as and when the program produces it.
- Actually need a datatype slightly more complicated than a list, to allow pure program to also receive *input* from the outside world.

A SimpleIO type constructor

- Simplified version of actual Haskell `IO` type constructor.
- Three-way choice:

```
data SimpleIO a = Done a
                | PutChar Char (SimpleIO a)
                | GetChar (Char -> SimpleIO a)
```

- Top-level loop has following conceptual structure:
  - If top-level expression has an “ordinary” (non-SimpleIO) type, just evaluate it and print the result (incrementally).
  - If expression has type `SimpleIO a`, evaluate it enough to expose top constructor:
    1. If of the form `Done` x, evaluate and print x, like in previous case
    2. If of the form `PutChar` c s, output c, and continue evaluating s.
    3. If of the form `GetChar` f, input a c, and continue evaluating $f\ c$.
- But how do we write a big program of type, say, `SimpleIO ()`?
  - Seems awkward to generate all IO requests in functional style.
  - Next time: monads to the rescue!

List comprehensions

- Cute Haskell feature, allows many list-processing functions to be written clearly and naturally.
- Inspired by mathematical notation for *set comprehensions*:
  - subset: $\{x \mid x \in \{2,3,5,7\} \wedge x > 4\} = \{5,7\}$
  - direct image: $\{x+1 \mid x \in \{2,3,5,7\}\} = \{3,4,6,8\}$
  - Cartesian product: $\{(x,y) \mid x \in \{2,3\} \wedge y \in \{\top,\bot\}\} = \{(2,\top),(2,\bot),(3,\top),(3,\bot)\}$
  - general union: $\{x \mid s \in \{\{2,3\},\emptyset,\{5\}\} \wedge x \in s\} = \{2,3,5\}$
- Can write Haskell expressions with almost same notation:
  - `[x | x <- [2,3,5,7], x > 4] == [5,7]`
  - `[x + 1 | x <- [2,3,5,7]] == [3,4,6,8]`
  - `[(x,y) | x <- [2,3], y <- [True,False]] == [(2,True), (2,False), (3,True), (3,False)]`
  - `[x | s <- [[2,3],[],[5]], x <- s] == [2,3,5]`

List comprehensions, continued

- Can even use all idioms on previous page together.

```
> [100 * x + y | x <- [1..4], x /= 3, y <- [1..x]]
[101,201,202,401,402,403,404]
```

- General shape: `[`exp `|` $qual_1, ..., qual_n$`]`, where each $qual_i$ is:
  - a *generator*, x `<-` $lexp_i$, where $lexp_i$ is a list-typed expression; or
  - a *guard*, $bexp_i$, which must be a `Bool`-typed expression.
- Qualifiers considered in sequence, from left to right:
  - For each generator, bind variable to successive list elements, and process next qualifiers ($\sim$ `foreach`-loop in imperative language)
  - For each guard, check that it evaluates to `True`; otherwise, return to previous generator ($\sim$ conditional `continue` in imperative).
  - When all qualifiers successfully considered, evaluate *exp* and add its value to result list.
- Aka. depth-first search, backtracking, generate-and-test
  - Will see again in Prolog, parsing
  - Also an instance of programming with monads!

What now?

- Talk to a fellow student about forming a group (two is max)
- Attend labs after lunch (rooms A1{01,02,03,07,10} at HCØ), from 12:30 (today only, then 12:45)
  - Don't need to come on time: no scheduled activities
  - Section and room assignments should be on Absalon; otherwise do ad-hoc load balancing.
- Work on Exercise Set 0
- Solve Assignment 0, **due 20:00 on Wednesday, 13 September**
  - Submission instructions Real Soon Now
  - Take advantage of OnlineTA: `find.incorrectness.dk`
- Use Absalon forum for questions after the lab hours
- Next lecture (**starting at 9:15**): monads!
  - Recommended reading materials will be up shortly.